
Changelog
    


  
    
  
  v0.6.1 (2026-04-22)


	Fix formatting of large app.extra output in mix uniform.eject (#49)


  
    
  
  v0.6.0 (2023-02-06)


	Adds support for suffix on eject fences (#46)


  
    
  
  v0.5.3 (2022-09-07)


	Fix to ensure Blueprint is recompiled when templates change (#45)


  
    
  
  v0.5.2 (2022-09-02)


	Include Blueprint template file in published Hex package (#43)


  
    
  
  v0.5.1 (2022-08-19)


	Fix issue where template files beginning with . weren't compiled (#40)
	Remove warning whenever app not used by a template (#40)


  
    
  
  v0.5.0 (2022-08-19)


	Support globs with Path.wildcard wildcard characters in file (#37)
	Remove support for lists of paths in file, template, cp, and cp_r (#37)


  
    
  
  v0.4.0 (2022-08-15)


	Rename code_fences to eject_fences
	Compile templates into Blueprint (#36)


  
    
  
  v0.3.0 (2022-08-09)


	Make templates optional in use Uniform.Blueprint (#33)
	Stop requiring # uniform:deps code fences in mix.exs (#34)


  
    
  
  v0.2.0 (2022-08-08)


	Change API of mix uniform.eject to accept my_app instead of MyApp (#31)
	Add mix uniform.ejectable_apps (#31)
	Add Uniform.ejectable_app_names and Uniform.ejectable_apps to public API (#31)
	Automatically import code_fences/3 in Blueprint (#30)


  
    
  
  v0.1.1 (2022-08-06)


	Support captured anonymous functions with modify/2 (#26)




  

    
Getting Started
    


  
    
  
  Add Uniform to mix.exs


Add :uniform as a dependency in mix.exs.
defp deps do
  [
    {:uniform, "~> 0.6.1"}
  ]
end

  
    
  
  Run mix uniform.init


After adding :uniform to mix.exs and running mix deps.get, run
mix uniform.init

This will Create a Blueprint module and add
Configuration, which you can do manually if you'd rather by
following the steps below.

  
    
  
  Create a Blueprint module


If you opted out of running mix uniform.init, create a
Blueprint module. This is the central file you'll use
to tell Uniform which files to copy when running mix uniform.eject.
To start, you can create an empty Blueprint like this.
defmodule MyApp.Uniform.Blueprint do
  use Uniform.Blueprint
end
You can name the module whatever you like, but we suggest putting it in
lib/my_app/uniform/blueprint.ex.

  
    
  
  Configuration


If you opted out of running mix uniform.init, add the following line to
config/config.exs. (Changing the blueprint value to match the name of your
Blueprint module above.)
config :my_app, Uniform, blueprint: MyApp.Uniform.Blueprint
You can also optionally set a default destination for ejected apps.
# With optional :destination
config :my_app, Uniform,
  blueprint: MyApp.Uniform.Blueprint,
  destination: "/Users/me/ejected"
If destination is ommitted, the default is one level up from the Base
Project's root folder. The --destination option of mix uniform.eject takes
precedence and overrides both of these behaviors.

  
    
  
  Add Uniform Manifests


Skip this step if you have no apps yet.
If you're setting up a brand new project, skip to Build the
Blueprint.
When you add Ejectable Apps in your
project, you'll need to come back to this step.

Put a uniform.exs file inside each directory in lib that represents an
Ejectable App. You can do so with this command:
mix uniform.gen.app my_app_name

It creates an empty manifest with code comments to help you start. If you want
to create the file manually instead, start with a barebones file that contains
an empty list:
# lib/my_app_name/uniform.exs
[]
The Uniform Manifests guide explains how
to use this file.
At this point you should be able to run:
mix uniform.eject my_app_name

And it will create a standalone codebase for the app. However, it probably
won't contain the files needed to run locally or be deployed.
This leads us to the final step of Building the Blueprint.

  
    
  
  Build the Blueprint


Read the documentation for Uniform.Blueprint for
the full range of features to build out your Blueprint module.

The Blueprint module is central for configuring which files are ejected during
mix uniform.eject.
Since each Elixir project is different, it is up to you to determine which
files need to be ejected in your project to emit a working application.
Are you building Phoenix apps?
The Setting up a Phoenix project guide
contains details and examples for building your Blueprint.

An example barebones Blueprint might look like this.
defmodule MyApp.Uniform.Blueprint do
  use Uniform.Blueprint

  base_files do
    file "lib/my_app/application.ex"
    file "lib/my_app_web/endpoint.ex"
    cp_r "assets"
    template "config/runtime.exs"
  end

  deps do
    always do
      mix :phoenix
      mix :phoenix_html

      lib :utilities
    end

    lib :my_data_graph do
      mix_deps [:absinthe]
    end
  end
end



  

    
How It Works
    

With Uniform, multiple apps are maintained together in a single Elixir
codebase. When you're ready to deploy an app, it's "ejected" to a separate
codebase that only contains the code needed by the app.

  
    
  
  The Base Project


The Base Project is the single Elixir project that houses multiple separate
apps. Apps are developed and tested together in the Base Project.
The Base Project's lib directory is central. It contains directories for:
	Ejectable Apps
	Lib Dependencies (shared libraries)

So the directory structure of a Base Project might look like this.
+ my_base_app
  + lib
    + my_first_app
    + my_second_app
    + utilities
    + ui_components

Each Base Project needs a Blueprint module
configuring which files are copied to ejected repositories.

  
    
  
  What is "Ejecting"?


"Ejecting" an app means copying the app's code to a separate, standalone
codebase – without including code the app doesn't need.
	Unused Lib Dependencies are excluded from lib
	Unused Mix Dependencies are removed from mix.exs

Ejecting is done with mix uniform.eject.
mix uniform.eject my_app_name


  
    
  
  Ejectable Apps


Ejectable Apps are apps that can be ejected from the Base
Project. Create them with mix uniform.gen.app.
mix uniform.gen.app my_new_app

To set up an Ejectable App manually:
	Make a directory inside lib for your app (E.g. lib/my_new_app)
	Add uniform.exs inside it


  
    
  
  Exactly which files get ejected?


There are four rules that determine which files are
copied during
ejection.
Basically, the Blueprint's base_files are ejected along with every directory
in lib and test whose name matches (1) the app being ejected and (2) its
Lib Dependencies.
Make sure to build out your Blueprint's
base_files and
deps sections. If you're building Phoenix
apps, you may want to consult the Setting up a Phoenix
project guide.



  

    
Uniform Manifests (uniform.exs)
    

As explained in the How It Works guide, Ejectable
Apps are defined by the existence of a
uniform.exs file inside the app's lib directory.
Each app has its own uniform.exs.

  
    
  
  Structure


uniform.exs contains a keyword list with three supported options: mix_deps,
lib_deps, and extra.
Each key is optional and defaults to [].
# the simplest valid uniform.exs
[]
A typical uniform.exs might look something like this.
# lib/my_app/uniform.exs
[
  mix_deps: [:gql, :timex],
  lib_deps: [:ui_components, :auth],
  extra: [
    some_data: "just for this app"
  ]
]

  
    
  
  mix_deps


mix_deps lists Mix Dependencies of the
app.
Provide the same atom as you do in mix.exs. (E.g. :ecto.)

  
    
  
  lib_deps


lib_deps lists Lib Dependencies of the
app.
Provide the directory in lib as an atom. (E.g. :ui_components for
lib/ui_components.)

  
    
  
  extra


extra contains a keyword list of arbitrary, developer-defined data. The
contents are placed in app.extra. (See Uniform.App)
When you need mix uniform.eject to emit different variations of code for
different apps beyond including/excluding dependencies, you'll probably want to
use app.extra with one of these tools:
	base_files
	Templates
	Modifiers

Note that app.extra also contains keys returned by the extra
callback. (uniform.exs has precedence for
conflicting keys.)



  

    
Dependencies
    


  
    
  
  Lib Dependencies


Sharing code between apps is central to the design of Uniform. Lib Dependencies
are the main code-sharing mechanism of Uniform.
A Lib Dependency is a code library in a subdirectory of the Base Project's
lib directory. For example:
	lib/auth, containing shared authentication code
	lib/ui, containing shared user interface code

Lib Dependencies are great for sharing non-public code between Ejectable Apps
with less ceremony than other mechanisms like private Hex packages.
You can easily add or remove Lib Dependencies
from ejected codebases as needed.

  
    
  
  Mix Dependencies


When an app is ejected, Uniform copies over the Base Project's mix.exs and
mix.lock, but removes unrequired mix deps. (See Adding Dependencies to an
App.)
To modify deps properly, Uniform requires a defp deps function in
mix.exs. This is the standard structure generated by mix new and mix phx.new.
defp deps do
  [
    ...
  ]
end

  
    
  
  Adding Dependencies to an App


By default, all dependencies are excluded from an app's ejected codebase. When you eject an app:
	Excluded Lib Dependencies will be missing from lib
	Excluded Mix Dependencies will be missing from mix.exs

There are three ways to include a dependency:
	Include a dependency in all Ejectable Apps by placing the it in the
always section of your Blueprint.

deps do
  always do
    lib :utilities
    mix :absinthe
  end
end
	Include a dependency in a single Ejectable App by saying so in
uniform.exs.

# lib/my_app/uniform.exs
[
  lib_deps: [:utilities],
  mix_deps: [:absinthe]
]
	Include transitive dependencies ("dependencies of dependencies") using the
lib_deps or
mix_deps macros. (See "Chained
Dependencies" below.)

deps do
  lib :some_included_lib do
    lib_deps [:utilities]
  end

  mix :some_included_mix do
    mix_deps [:absinthe]
  end
end

  
    
  
  Chained Dependencies


mix uniform.eject follows chains of sub-dependencies completely.
Imagine this scenario.
	uniform.exs has lib_deps: [:foo]
	The Blueprint's deps says foo has lib_deps: [:bar]
	The Blueprint's deps says bar has lib_deps: [:baz]

As a result, the ejected codebase will include lib/foo, lib/bar, and
lib/baz.



  

    
Code Transformations
    

During mix uniform.eject, there are 4 code transformations applied to file
contents. These transformations happen to every file except those ejected with
cp and cp_r.
They occur in this order.
	Unused mix.exs Dependencies are Removed
	Blueprint Modifiers are ran
	The Base Project Name is replaced with the ejected app's name
	Eject Fences are processed


  
    
  
  mix.exs Dependency Removal


Any Mix Dependency that is not directly or indirectly required by the app via
uniform.exs or the Blueprint module is removed from the ejected mix.exs.

  
    
  
  Blueprint Modifiers


Users can specify arbitrary modifications that should be applied to various
files using the modify macro in the Blueprint module:
modify ~r/.+_worker.ex/, fn file, app ->
  # `file` is a string containing the full file contents.
  # `app` is the `Uniform.App` struct. (The app being ejected.)
  # The string this function returns will be the ejected file contents.
end

modify "lib/my_app_web/router.ex", fn file ->
  # This modifier is like the one above, but the transformation will only
  # be ran for `lib/my_app_web/router.ex`.
end

  
    
  
  Replacing the Base Project Name


The base project name, appearing anywhere in a file, is replaced by the ejected
app name. This applies to the following formats: base_app, base-app, and
BaseApp.
The base project name is the :app key returned by project in the mix.exs
file of the Base Project. (For example, :my_base_app below.)
# mix.exs
def project do
  [
    app: :my_base_app, # <- base project name
    ...
  ]
end
Given the above mix.exs, if you were to run mix uniform.eject my_ejectable_app:
	my_base_app would be replaced with my_ejectable_app
	my-base-app would be replaced with my-ejectable-app
	MyBaseApp would be replaced with MyEjectableApp

Replacement in file paths
This same replacement of base_project_name to ejected_app_name also
occurs in file paths, but only with this_format. (Not this-format or
ThisFormat.)
This means a file at lib/base_project_name/foo.ex would be ejected to
lib/ejected_app_name/foo.ex.

This means that a file like this
defmodule MyBaseAppWeb.Endpoint do
  use Phoenix.Endpoint, otp_app: :my_base_app

  socket "/socket", MyBaseAppWeb.UserSocket,
    websocket: true,
    longpoll: false

  plug MyBaseAppWeb.Router
end
Would be transformed to this
defmodule MyEjectableAppWeb.Endpoint do
  use Phoenix.Endpoint, otp_app: :my_ejectable_app

  socket "/socket", MyEjectableAppWeb.UserSocket,
    websocket: true,
    longpoll: false

  plug MyEjectableAppWeb.Router
end

  
    
  
  Eject Fences


In any .ex or .exs file, you can use "Eject Fence" comments to remove code
unless certain criteria are met.
To remove code unless the ejected app depends on a Lib Dependency called
my_lib, wrap it in these comments:
# uniform:lib:my_lib
# ... code
# /uniform:lib:my_lib
To remove code unless the ejected app depends on a Mix Dependency called
absinthe, wrap it in these comments:
# uniform:mix:absinthe
# ... code
# /uniform:mix:absinthe
To remove code unless the ejected app is called MyApp, wrap it in these
comments:
# uniform:app:my_app
# ... code
# /uniform:app:my_app
Finally, to always remove a chunk of code whenever ejection happens, wrap
it in these comments:
# uniform:remove
# ... code
# /uniform:remove
Eject Fence comments are always removed
Regardless of whether mix uniform.eject keeps or deletes the code in an
eject fence, the eject fence comments themselves (like # uniform:app:my_app)
are always removed.
Furthermore, mix uniform.eject runs mix format on the ejected codebase at
the end. So you always end up with "clean" looking code.


  
    
  
  Eject Fences for other languages


Eject Fences are also processed for .js/.ts/.jsx/.tsx files using JS
single-line comments.
// uniform:lib:my_lib
// ...
// /uniform:lib:my_lib
If you would like to support Eject Fences for other languages or file types, you
can do so using Uniform.Modifiers.eject_fences/3.
# eject fences for SQL files
modify ~r/\.sql$/, fn file, app ->
  eject_fences(file, app, "--")
end

# eject fences for Rust files
modify ~r/\.rs$/, &eject_fences(&1, &2, "//")

# eject fences for CSS files
modify ~r/\.css$/, &eject_fences(&1, &2, "/\\*", "\\*/")

  
    
  
  Disabling Code Transformations


If you have a file that should not have Code Transformations applied upon
ejection, use cp instead of
file.
If there is an entire directory of contents that should not be modified, use
cp_r, which will be much faster.



  

    
Setting up a Phoenix project
    

This guide walks through a typical process for setting up a Phoenix project.
It assumes you've read the How It Works and Getting
Started guides.

Imagine creating an entire new application simply by adding a Route and a
single LiveView to an existing Elixir project.
This is in fact the vision that drove the creation of Uniform. The overhead of
experimenting with new apps becomes extremely low, incentivizing the team to
try out ideas without being inhibited by initial setup time.

  
    
  
  Setting up your Blueprint module


Below is an example Blueprint module for ejecting the files
common to Phoenix applications.
defmodule MyBaseApp.Uniform.Blueprint do
  use Uniform.Blueprint, templates: "lib/my_base_app/uniform/templates"

  base_files do
    cp_r "assets"
    cp_r "priv/static"

    file "lib/my_base_app_web.ex"
    file "config/**/*.exs"
    file "test/support/*_case.ex"
  end

  deps do
    # Always eject the `my_base_app` and `my_base_app_web` libraries.
    #
    # Their paths and file contents will replace my_base_app with the ejected
    # app name automatically.
    always do
      lib :my_base_app do
        only [
          "lib/my_base_app/application.ex",
          "lib/my_base_app/repo.ex"
        ]

        # `match_dot: true` so that we eject `priv/repo/.formatter.exs`
        file "priv/repo/**/*.exs", match_dot: true
      end

      lib :my_base_app_web do
        only [
          "lib/my_base_app_web/endpoint.ex",
          "lib/my_base_app_web/gettext.ex",
          "lib/my_base_app_web/router.ex",
          "lib/my_base_app_web/telemetry.ex",
          "lib/my_base_app_web/channels/user_socket.ex",
          "lib/my_base_app_web/templates/layout/app.html.heex",
          "lib/my_base_app_web/templates/layout/live.html.heex",
          "lib/my_base_app_web/templates/layout/root.html.heex",
          "lib/my_base_app_web/views/error_helpers.ex",
          "lib/my_base_app_web/views/error_view.ex",
          "lib/my_base_app_web/views/layout_view.ex"
        ]
      end
    end
  end
end
Let's walk through it step by step.

  
    
  
  The base_files section


base_files do
  cp_r "assets"
  cp_r "priv/static"

  file "lib/my_base_app_web.ex"
  file "config/**/*.exs"
  file "test/support/*_case.ex"
end
In the base_files section, we specify
files that should be ejected in every app. Phoenix apps will typically have
CSS and JS assets in the assets directory. They'll also have static files to
be served as-is in priv/static. Most of these files will typically be binary
(non-text) files. So we assume none of them need to pass through the Code
Transformation phase. That's why the first two
lines are included.
cp_r "assets"
cp_r "priv/static"
Note that cp_r instructs mix uniform.eject to copy all the directory
contents (using File.cp_r!/3).
Phoenix apps typically have a Web module which is used to construct
Controllers, Views, Routers, and LiveViews. Since this file is typically in
lib/ directly (and not in a sub-directory of lib/), we include it here in
the base_files section.
file "lib/my_base_app_web.ex"
For the last two lines below, we'll use wildcard
characters to target multiple
files. (file accepts a Path.wildcard/2 glob or a concrete path.)
We proceed with the assumption that the ejected app will need the Base
Project's configuration files.
file "config/**/*.exs"
Lastly, we include all the custom test cases in the Base Project.
file "test/support/*_case.ex"

  
    
  
  Using front-end JavaScript Libraries


If you're using a front-end JavaScript library like React.js, you probably
don't want to eject all of the contents of assets with every app like this.
Depending on your setup, you may want to put JS files in separate directories
for each app and include them in base_files in one of these ways.
# with Code Transformations
base_files do
  file "assets/#{app.name.underscore}/**/*.{js,ts}"
end
# without Code Transformations
base_files do
  cp_r "assets/#{app.name.underscore}"
end

  
    
  
  The deps section


In the deps section, we put both lib :my_base_app and lib :my_base_app_web inside always do so that their contents are always ejected
without having to specify lib_deps: [:my_base_app, :my_base_app_web] in the
uniform.exs manifest of every app.
deps do
  always do
    lib :my_base_app do
      # ...
    end

    lib :my_base_app_web do
      # ...
    end
  end
end
For :my_base_app, we use an only instruction to exclude all files in
lib/my_base_app and test/my_base_app except for application.ex and
repo.ex.
lib :my_base_app do
  only [
    "lib/my_base_app/application.ex",
    "lib/my_base_app/repo.ex"
  ]

  file "priv/repo/**/*.exs", match_dot: true
end
You may have a setup that requires you to add more files, such as mailer.ex.
Note that we also include all of the Repo's migrations and seeds scripts with
a glob: priv/repo/**/*.exs. Both that and the match_dot option are passed
to Path.wildcard/2 under the hood. match_dot: true ensures that
priv/repo/.formatter.exs is ejected so that the ejected codebase is formatted
properly.
For :my_base_app_web, we also use an only instruction to only include
relevant files.
lib :my_base_app_web do
  only [
    "lib/my_base_app_web/endpoint.ex",
    "lib/my_base_app_web/gettext.ex",
    "lib/my_base_app_web/router.ex",
    "lib/my_base_app_web/telemetry.ex",
    "lib/my_base_app_web/channels/user_socket.ex",
    "lib/my_base_app_web/templates/layout/app.html.heex",
    "lib/my_base_app_web/templates/layout/live.html.heex",
    "lib/my_base_app_web/templates/layout/root.html.heex",
    "lib/my_base_app_web/views/error_helpers.ex",
    "lib/my_base_app_web/views/error_view.ex",
    "lib/my_base_app_web/views/layout_view.ex"
  ]
end

  
    
  
  The Phoenix Router


A simple way to set up your Phoenix.Router is to put the routes for all of
your apps in a single router. Then, use Eject
Fences and Uniform.Blueprint.modify/2
to transform the router upon ejection.
Let's look at an example. We'll explain each part below.
defmodule MyBaseAppWeb.Router do
  use MyBaseAppWeb, :router

  pipeline :browser do
    # ...
  end

  # uniform:app:some_app
  scope "/some-app", SomeAppWeb do
    pipe_through :browser

    get "/widgets", WidgetController, :index
    get "/widgets/new", WidgetController, :new
    post "/widgets/new", WidgetController, :create
    get "/widgets/:widget_id", WidgetController, :show
  end
  # /uniform:app:some_app

  # uniform:app:another_app
  scope "/another-app", AnotherAppWeb do
    pipe_through :browser

    get "/posts", PostController, :index
    get "/posts/new", PostController, :new
    post "/posts/new", PostController, :create
    get "/posts/:post_id", PostController, :show
  end
  # /uniform:app:another_app
end
You'll want to structure the router for reuse by including any pipelines (e.g.
:browser or :api) that your apps will need.
pipeline :browser do
  # ...
end
Next, add scopes for each app, wrapped in Eject
Fences. This ensures ejected routers
will only contain routes related to the ejected app.
# uniform:app:some_app
scope "/some-app", SomeAppWeb do
  # ...
end
# /uniform:app:some_app
You'll need to add controllers/views/etc inside of the app's lib directory.
For example, SomeAppWeb.WidgetController should be in lib/some_app.
(Probably at lib/some_app/controllers/widget_controller.ex.)

Prefix the paths of each scope with /app-name (like /some-app above) so
that each app has a predictable, separated URL structure when running the Base
Project locally.
As a last step, we need to remove the /app-name path prefix
during ejection. We can do this with modify.
defmodule MyBaseApp.Uniform.Blueprint do
  use Uniform.Blueprint

  modify "lib/my_base_app_web/router.ex", fn file, app ->
    String.replace(
      file,
      "scope \"/#{app.name.hyphen}\"",
      "scope \"/\""
    )
  end
end
With this modifier, the following code
scope "/some-app", SomeAppWeb do
Changes to
scope "/", SomeAppWeb do
In the ejected codebase.
This method is a great starting point. Before you reach dozens of apps, you may
want to consider other methods that allow you to define routes in a separate
file per app.

  
    
  
  Internal Pages


We encourage running all apps simultaneously via the Base Project as your
development environment. In such a setup, it can be useful to add other pages
to the Base Project that aren't intended to be ejected with any app.
For example, you might add a page that catalogs and links to your various apps.
We recommend adding these routes and wrapping them all in # uniform:remove
Eject Fences as in the example above.
# uniform:remove
scope "/", SomeAppWeb do
  pipe_through :browser

  get "/internal-team-page", InternalTeamController, :index
end
# /uniform:remove

  
    
  
  Eject Fences Everywhere!


There are other files which are central for running Elixir apps.
Similarly to the Phoenix Router, we recommend that you add the code required by
each of your apps and Lib Dependencies to
all of these files. Then, use Eject Fences
to selectively remove code during ejection.
Let's examine what this might look like for application.ex, mix.exs, and
config/*.exs files.

  
    
  
  Application


Your Application file at lib/my_base_app/application.ex is a critical piece
of Elixir applications since it's used to start processes and supervisors at
the start of the application.
Here's what an example Application file would look like with Eject Fences
applied.
defmodule MyBaseApp.Application do
  use Application

  def start(_type, _args) do
    children = [
      MyBaseAppWeb.Endpoint,
      {Phoenix.PubSub, name: MyBaseApp.PubSub},
      MyBaseAppWeb.Presence,
      MyBaseAppWeb.Telemetry,

      # uniform:lib:my_first_data_lib
      MyFirstDataLib.Repo,
      # /uniform:lib:my_first_data_lib

      # uniform:lib:my_second_data_lib
      MySecondDataLib.Repo,
      MySecondDataLib.Vault,
      # /uniform:lib:my_second_data_lib

      # uniform:remove
      SomeDevelopmentOnlyDB.Repo,
      # /uniform:remove

      # uniform:mix:oban
      {Oban, ...},
      # /uniform:mix:oban
    ]

    # ...
  end
end
Notice that code which should always be ejected does not get surrounded by
Eject Fences.

  
    
  
  mix.exs


Some dependencies require modifying mix.exs. For example, the
exq Hex package says to add :exq to application
in mix.exs.
def application do
  [
    applications: [:logger, :exq],
    # ...
  ]
end
But what if only some of your apps require exq? Wrap the exq-specific code in
Eject Fences, and it will only be included when exq is required.
def application do
  [
    applications: [
      :logger,
      # uniform:mix:exq
      :exq
      # /uniform:mix:exq
    ],
    # ...
  ]
end
You don't need to use Eject Fences in deps
Note that removing deps from the deps section of mix.exs is automatic, so
this would not be required.
# ❌ There is no need to wrap individual deps in eject fences
defp deps do
  [
    # uniform:mix:jason
    {:jason, "~> 1.0"}
    # /uniform:mix:jason
  ]
end

  
    
  
  Config Files


Many dependencies also require configuration. Apply Eject Fences in your
configuration files for the same result.
# uniform:mix:guardian
config :my_base_app, MyBaseApp.Guardian,
       issuer: "my_base_app",
       secret_key: ...
# /uniform:mix:guardian



  

    
Auto-updating ejected codebases
    

When using Uniform, we recommend the following setup:
	Host ejected code repositories on the same platform as your Base
Project. (GitHub, GitLab, etc.)
	Consider those ejected repositories as "untouchable". Only update them via
mix uniform.eject.
	Automate updating ejected repositories with Continuous Integration (CI).


  
    
  
  Building your CI pipeline


If you follow the setup recommended above, you'll need to build a Continuous
Integration (CI) pipeline that commits updates to your ejected code
repositories.
Since each CI tool is configured in a different way, we cannot provide a single
example to match your platform. However, we can provide a list of recommended
steps to build the pipeline.
In your CI pipeline, we recommend:
	Only eject after a successful merge to the main/master branch.
	Use mix uniform.ejectable_apps to get a full list of Ejectable Apps. (See below.)
	Clone each of your Ejectable Apps so that they're ready to have commits
added.
	For each app, run mix uniform.eject.
	For each app, run git commit. If there are no changes to commit, this will
no-op.
	Push your changes with git push.

Note that these steps leave out whatever authentication steps are necessary
with your git host to clone and push to the repositories. Since each git
host is different, you'll have to consult the documentation for yours.

  
    
  
  Getting a list of Ejectable Apps


For step 2 above, Uniform ships with a few tools to help you get a full list of
your Ejectable Apps.
To output your Ejectable App names in stdout on the command line, use mix uniform.ejectable_apps.
$ mix uniform.ejectable_apps
my_first_app
my_second_app

If you're working in Elixir, you can also use Uniform.ejectable_app_names/0
or Uniform.ejectable_apps/0.



  

    
Handling multiple data sources
    

This guide assumes that you've read the How It Works and
Getting Started guides.

If you're using Uniform it's likely that you have multiple databases and/or
remote APIs that you query, but you don't want to give every app access to
every one.
In this scenario, we recommend creating separate Lib
Dependencies (each in its own
lib/some_data_source directory) which each encapsulate all of the code for
interacting with a single database or API. This implies that each Lib
Dependency would house the Ecto Repo
as well as all Ecto Schemas and
Context modules related to its
database.
Structuring the code this way allows you to easily include or exclude the code
for a data source in uniform.exs.

  
    
  
  An Example


For example, given apps with these uniform.exs manifests
# lib/some_app/uniform.exs
[
  lib_deps: [:my_data_source]
]
# lib/another_app/uniform.exs
[
  lib_deps: [:my_data_source, :other_data_source]
]
# lib/third_app/uniform.exs
[
  lib_deps: [:other_data_source]
]
	SomeApp would be ejected with all of the code related to interacting with
my_data_source
	AnotherApp would be ejected with the code for both my_data_source and
other_data_source
	ThirdApp would only be ejected with the code for other_data_source


  
    
  
  Configuring Lib Dependencies


If you take this approach, make sure to configure your
Blueprint module to include any Lib or Mix Dependencies
of each data source's library. Also, be sure to include migrations and seeds
related to the library.
lib :my_data_source do
  lib_deps [:some_library]
  mix_deps [:faker, ...]

  # `match_dot: true` below to include priv/my_data_source_repo/.formatter.exs
  file "priv/my_data_source_repo/**", match_dot: true
end



  

    
Building files from EEx templates
    

EEx templates can be used to create files in Ejected Apps that don't exist in
the Base Project.

  
    
  
  An Example Scenario


Imagine you do not use runtime
configuration
in your Base Project, but you want a config/runtime.exs in your ejected apps.
Furthermore, imagine the contents of runtime.exs need to be wildly different
between apps based on the Lib and Mix Dependencies of the app.
This is an ideal scenario for using an EEx
template. Below are the steps to create runtime.exs with an EEx template.

  
    
  
  Step 1: Set the template directory in your Blueprint module


defmodule MyApp.Uniform.Blueprint do
  use Uniform.Blueprint, templates: "lib/uniform/templates"

  
    
  
  Step 2: Add the EEx file in the correct relative path


EEx templates must be added in the destination path, relative to the
templates directory from use Uniform.Blueprint, with a .eex suffix.
So to emit our file into config/runtime.exs, we must create it in
lib/uniform/templates/config/runtime.exs.eex
\___________________/ \________________/\__/
        |                     |           |
templates directory     destination    .eex suffix (Don't forget!)


  
    
  
  Step 3: Write the template


Here is an example of what the runtime.exs.eex template might contain.
Note the use of Eject Fences at the
end of the file.
import Config

<%= if MyApp.Uniform.Blueprint.deploys_to_fly_io?(app) do %>
  config :my_base_app, some_api_token: System.get_env("SOME_API_TOKEN")
<% end %>

config :my_base_app, MyBaseAppWeb.Endpoint,
  url: [
    host: System.get_env("HOST"),
    port: String.to_integer(System.get_env("EXTERNAL_PORT"))
  ],
  secret_key_base: System.get_env("SECRET_KEY_BASE")

config :my_base_app, MyBaseApp.Repo,
  priv: "priv/my_base_app_repo",
  url: System.get_env("DATABASE_URL")

# uniform:app:some_app
config :my_base_app, some_configuration: "just for some_app"
# /uniform:app:some_app
Keep in mind a few considerations documented below when building out templates.

  
    
  
  The App struct


In EEx templates, the Uniform.App struct will be available as app. You can
use the contents of app.extra to make decisions about what to render in the
template.
Also, the Uniform.App.depends_on?/3 utility is available as depends_on? and
can be used like this.
depends_on?(app, :mix, :absinthe)

  
    
  
  Code transformations


Code Transformations are ran after the
template is generated. This means that you can inject the base app name into a
template anywhere you want the ejected app name to appear, and it will be
transformed into the ejected app name.
So in the example above, instead of writing
config :<%= app.name.underscore %>
We just wrote
config :my_base_app



  

    
Use Cases
    


  
    
  
  From a High Level


Uniform is powerful in scenarios in which you want to:
	Maintain a portfolio of apps
	Deploy them separately
	Keep many aspects of them in sync
	Share capabilities between apps

This makes Uniform especially powerful for software organizations that do not
have an entire team dedicated to each app.

  
    
  
  Enterprises


In our specific use case, we build and maintain apps across companies for a
large enterprise. The companies have their own distinct branding, themes,
deployment pipelines, and assurance processes. However, they share data access
patterns, UI libraries, authentication mechanisms, and many other common
features.
By leveraging Uniform, we're able to perform a change to our entire portfolio
of apps in the same time that it would normally take to make the change to a
single app. For example, when we find a security issue in our shared code
architecture, we patch it once and all apps are instantly updated. What could
have been a lengthy process to audit and patch dozens of apps becomes much
simpler and less time-intensive.
Uniform also helps us accomplish sharing capabilities among apps. While
building a feature for a single app, we might create an "autosuggest" UI
component for suggesting live results as the user types in an input field. With
extremely little effort, by simply moving the component out of lib/that_app
into lib/shared_ui_components, this component becomes instantly available in
all other apps.

  
    
  
  Agencies


Agencies that build custom apps often find themselves pressed for time as they
try to balance the client's financial constraints with feature wishlists and
the health of the codebase.
Uniform can help an agency leverage as much value as possible from their
portfolio of already-built apps as they start new projects. Instead of
rebuilding core pieces like project boilerplate, authentication systems, or
integrations with third party services, those capabilities are available from
the start. (As long as they're built with sharing in mind.)
Furthermore, since Uniform emits standalone codebases for each app, an agency
is able to leverage this extreme level of reuse while still having separate
codebases that belong to each client.



  

    
Benefits and Disadvantages
    

Uniform is not for everyone, but it can be very powerful in environments where
a team or developer is tasked with developing a large portfolio of apps that
have many similarities or could share many capabilities.

  
    
  
  Benefits


	Higher leverage to maintain a portfolio of apps. There is a sense of
higher return on investment (ROI) as more and more apps are built. When you
add a new capability, refactor deprecations, or change an implementation, the
results are automatically incorporated into all of your apps.
	Reduce dependency drift. Ejected apps derive their Mix and NPM
dependencies from the Base Project. This eliminates distractions that come
from supporting multiple versions across a portfolio. No more grappling with
missing features or incompatible APIs.
	Faster feedback. When you change anything shared, your changes are
automatically made and tested in all apps that could be impacted.
	Share capabilities with lower overhead. Any capability that is developed
for a specific app can be moved into a Lib Dependency. At that point, it's
instantly available to all other apps in the portfolio.
	Defer factoring decisions. In the interest of the DRY (Don't Repeat
Yourself) principle, there is a motivation to abstract and modularize the
moment we start adding a feature. Many advocate for the WET principle (Write
Everything Twice) by contrast, since making abstractions too early can often
result in an abstraction that "leaks" or is otherwise ill-suited to the
specific use case. Since Uniform lowers the amount of overhead for shared
internal libraries, there are no distinct packages to create up front which
makes it much easier to refactor as the API evolves.
	Global Refactors with peace of mind. When the entire portfolio of
apps is managed together, we're free to refactor any interal API and know
that we've addressed all invocations across the enterprise. With Elixir
compiler warnings and integration tests, CI builds reveal invocations we fail
to update. In many cases, this eliminates the burden of maintaining backwards
compatibility, leading to simpler internal libraries.
	"Release" an ejected app at any time. If at any point you need to release
control of an ejected app to another entity such as a different
development agency, the ejected codebase is already ready as a standalone
project.
	Self-healing properties. - Since mix uniform.eject starts by deleting
the contents of the destination codebase, it's possible to modify an ejected
codebase temporarily, and then "bring it back in sync" with your Base Project
by running mix uniform.eject again. Ejecting an app with mix uniform.eject is meant to be something that can be done over and over
without preparation or ceremony.


  
    
  
  Disadvantages


	Novelty. When new developers join the team, contributors have to learn a
new way of working.
	No contributing to ejected codebases. Changes made directly to an ejected
codebase cannot be automatically integrated back into the Base Project, so
contributors must have access to the entire portfolio.
	Complications from running multiple apps at once. Your development
environment is more complicated when you're running an entire suite of apps
side-by-side. Teams are required to take a disciplined approach to deal with
some of these complications.	For example, the paths of routes would typically be prefixed with the app
name (e.g. localhost:4000/my-app/users instead of
localhost:4000/users), so that paths in your dev environment don't
exactly match production. One implication is that if you write links like
this <a href="/users"> instead of like this <a href="<%= Routes.users_path(...) %>"> the link will work in development but not
production.


	Compilation time. Full recompiles can take a long time for a large
portfolio of apps. Elixir 1.13 has helped out tremendously in this
regard. On the other hand, you'll often end up with a full recompile when
changing something in config/config.exs which only has relevance to a
single app.
	CI burn rate. We run Continuous Integration (CI) checks for every ejected
app whenever a commit is made to the Base Project. We suggest you do, too.
However, this can burn through an order of magnitude more CI minutes.




  

    
Uniform 
    



      
Write less boilerplate and reuse more code in your portfolio of Elixir apps

With Uniform, developers maintain multiple Elixir apps inside a Base Project:
a "monolith" containing every app. Before deployment, the apps are "ejected"
into separate codebases containing only the code needed by each app.
The entire process is automated, so there's much less work required to start
a new app or share capabilities between apps.

  
    
  
  Recommended Guides


In order to understand and use this library, we heavily recommend reading the
following guides:
	How It Works
	Uniform Manifests
	Dependencies
	Code Transformations

The Setting up a Phoenix project guide
is recommended if you're building Phoenix apps.

  
    
  
  Usage


mix uniform.eject tweeter


  
    
  
  Installation


Consult the Getting Started guide to add Uniform to
an Elixir application. In summary, you'll need to:
	Add the dep in mix.exs
	Add a Blueprint module to your project
	Add configuration to identify the Blueprint module
	Add uniform.exs manifests to each Ejectable App
	Fill out the Blueprint so all the necessary files get ejected

Uniform ❤️ Automation
Uniform is about powering up developers by automating repetitive work.
With that in mind, we recommend using Continuous Integration (CI) tools to
automate the process of committing code to ejected
repos.


      


      
        Summary


  
    Functions
  


    
      
        ejectable_app_names()

      


        Returns a list of all Ejectable App names
in your Base Project.



    


    
      
        ejectable_apps()

      


        Return a list of all Ejectable Apps in
your Base Project as Uniform.App structs.



    





      


      
        Functions

        


  
    
      
      Link to this function
    
    ejectable_app_names()


      
       
       View Source
     


  


  

      

          @spec ejectable_app_names() :: [String.t()]


      


Returns a list of all Ejectable App names
in your Base Project.

  
    
  
  Examples


$ fd uniform.exs
lib/tweeter/uniform.exs
lib/trillo/uniform.exs
lib/hatmail/uniform.exs

iex> ejectable_app_names()
["tweeter", "trillo", "hatmail"]

  



  
    
      
      Link to this function
    
    ejectable_apps()


      
       
       View Source
     


  


  

      

          @spec ejectable_apps() :: [Uniform.App.t()]


      


Return a list of all Ejectable Apps in
your Base Project as Uniform.App structs.

  
    
  
  Example


$ fd uniform.exs
lib/tweeter/uniform.exs
lib/trillo/uniform.exs
lib/hatmail/uniform.exs

iex> ejectable_apps()
[
  #Uniform.App<
    extra: [...],
    name: %{camel: "Tweeter", hyphen: "tweeter", module: Tweeter, underscore: "tweeter"},
    ...
  >,
  #Uniform.App<
    extra: [...],
    name: %{camel: "Trillo", hyphen: "trillo", module: Trillo, underscore: "trillo"},
    ...
  >,
  #Uniform.App<
    extra: [...],
    name: %{camel: "Hatmail", hyphen: "hatmail", module: Hatmail, underscore: "hatmail"},
    ...
  >
]

  


        

      



  

    
Uniform.App 
    



      
A struct representing an app to be ejected. See the type
definition for more details.

  
    
  
  Where it's availabile


An app struct is available in the Blueprint in these callbacks and macros:
	base_files
	modify
	extra
	target_path

Use the app to make decisions about what to eject or how files should be
modified.

  
    
  
  Checking for dependencies


In the callbacks and macros above, you can use depends_on? to determine
whether an app depends on a mix or lib dependency.
depends_on?(app, :mix, :norm)
See depends_on?/3 for more information.

      


      
        Summary


  
    Types
  


    
      
        t()

      


        An App struct, representing a discrete, self-contained app to be ejected.



    





  
    Functions
  


    
      
        depends_on?(app, category, dep_name)

      


        Indicates if an app requires a given dependency.



    





      


      
        Types

        


  
    
      
      Link to this type
    
    t()


      
       
       View Source
     


  


  

      

          @type t() :: %Uniform.App{
  destination: Path.t(),
  extra: keyword(),
  internal: term(),
  name: %{
    module: module(),
    hyphen: String.t(),
    underscore: String.t(),
    camel: String.t()
  }
}


      


An App struct, representing a discrete, self-contained app to be ejected.

  
    
  
  Example


Note that the extra key contains everything you put in extra in
uniform.exs for the given app. It also contains anything returned by
Uniform.Blueprint.extra/1. (uniform.exs has precedence for conflicting
keys.)
#Uniform.App<
  extra: [
    company: :fake_co,
    logo_file: "path/to/some_logo.png",
    some_data: "from uniform.exs"
  ],
  name: %{
    camel: "Tweeter",
    hyphen: "tweeter",
    module: Tweeter,
    underscore: "tweeter"
  },
  ...
>

  


        

      

      
        Functions

        


  
    
      
      Link to this function
    
    depends_on?(app, category, dep_name)


      
       
       View Source
     


  


  

      

          @spec depends_on?(app :: t(), category :: :lib | :mix, dep_name :: atom()) ::
  boolean()


      


Indicates if an app requires a given dependency.
Pass in the app, the dependency type (either :lib or :mix), and the
name of the dependency (like :tesla or :my_lib_directory) and the
function will return true if the dependency will be ejected along with the
app.

  
    
  
  Examples


depends_on?(app, :mix, :some_included_mix_dep)
depends_on?(app, :mix, :not_included_dep)
depends_on?(app, :lib, :some_included_lib)

  
    
  
  Examples in Context


base_files do
  if depends_on?(app, :mix, :some_hex_dependency) do
    file "file_needed_by_some_hex_dependency"
  end
end

modify ~r/^test/.+_(test).exs/, fn file, app ->
  if depends_on?(app, :lib, :my_data_lib) do
    file
  else
    String.replace(
      file,
      "use Oban.Testing, repo: MyDataLib.Repo",
      "use Oban.Testing, repo: OtherDataLib.Repo"
    )
  end
end

  


        

      



  

    
Uniform.Blueprint behaviour
    



      
Defines the ejection blueprint for your project.
When used, the blueprint accepts the :templates option, which is not
required. For example, the blueprint:
defmodule Blueprint do
  use Uniform.Blueprint, templates: "priv/uniform-templates"
end
Would search for templates in the priv/uniform-templates directory. See
template/2 for more information.

  
    
  
  The base_files Section


The base_files section specifies files that should be ejected which aren't
in lib/my_app. (When running mix uniform.eject my_app.)
defmodule Blueprint do
  use Uniform.Blueprint

  base_files do
    file "my_main_app/application.ex"
    cp_r "assets"
    # ...
  end
end
See base_files/1 for more information.

  
    
  
  Files that are always ejected


There are a handful of files that are automatically ejected. You do not need
to specify these in the base_files section.
mix.exs
mix.lock
.gitignore
.formatter.exs
test/test_helper.exs


  
    
  
  The deps Section


Besides the base_files section, the blueprint can also contain a deps
section to configure dependencies.
defmodule Blueprint do
  use Uniform.Blueprint

  deps do
    always do
      mix :phoenix
      lib :my_component_library
    end

    mix :absinthe do
      mix_deps [:absinthe_plug, :dataloader]
    end
  end
end
See deps/1 for more information.

  
    
  
  Modifying files programmatically with modify


Lastly, modify can be used whenever you want to transform file contents
during ejection. You can specify a specific filepath or use a regex to match
multiple files.
defmodule Blueprint do
  use Uniform.Blueprint

  modify "assets/js/app.js", fn file, app ->
    String.replace(file, "SOME_VALUE_PER_APP", app.extra[:some_value])
  end

  modify ~r/_worker.ex/, &MyApp.Modify.modify_workers/1
See modify/2 for more information.

  
    
  
  Preserving files


Before mix uniform.eject copies any files, the contents in the destination
directory are deleted – except for the .git, deps, and _build
directories.
If there are any other files or directories in the project's root folder
that you would like to preserve (by not deleting them), specify them with
@preserve.
# .env will not be deleted immediately before ejection
@preserve [
  ".env"
]

  
    
  
  Full Example


Below is an example Blueprint module that shows off a majority of the
features that can be used.
defmodule MyApp.Uniform.Blueprint do
  use Uniform.Blueprint, templates: "lib/uniform/templates"

  # do not delete this root file when clearing the destination
  @preserve [".env"]

  @impl Uniform.Blueprint
  def extra(app) do
    theme =
      case app.name.underscore do
        "empire_" <> _ -> :empire
        "rebel_" <> _ -> :rebel
      end

    # set `app.extra[:theme]`
    [theme: theme]
  end

  @impl Uniform.Blueprint
  def target_path(path, app) do
    if is_web_file?(path) do
      # modify the path to put it in `lib/some_app_web`
      String.replace(path, "lib/#{app.name.underscore}/", "lib/#{app.name.underscore}_web/")
    else
      path
    end
  end

  # files to eject in every app, which are outside `lib/that_app`
  base_files do
    # copy these directories wholesale; do NOT run them through code modifiers
    cp_r "assets"

    # eject these files which aren't in lib/the_app_directory
    file ".credo.exs"
    file ".github/workflows/elixir.yml"
    file "priv/static/#{app.extra[:theme]}-favicon.ico"
    file "lib/my_app_web.ex"

    # eject a file from an EEx template at "lib/uniform/templates/config/runtime.exs.eex"
    # configure the templates directory on line 2
    template "config/runtime.exs"

    # conditionally eject some files
    if deploys_to_aws?(app) do
      file "file/required/by/aws"
      template "dynamic/file/required/by/aws"
    end

    if depends_on?(app, :lib, :some_lib) do
      template "dynamic/file/required/by/some_lib"
    end
  end

  # run the file contents through this modifier if the file is ejected
  modify "lib/my_app_web/templates/layout/root.html.heex", file, app do
    file
    |> String.replace("empire-favicon.ico", "#{app.extra[:theme]}-favicon.ico")
    |> String.replace("empire-apple-touch-icon.png", "#{app.extra[:theme]}-apple-touch-icon.png")
  end

  # configure dependencies from mix.exs and `lib/`
  deps do
    # always eject the dependencies in the `always` section;
    # don't require adding them to uniform.exs
    always do
      lib :my_app do
        # only eject the following files in `lib/my_app`
        only ["lib/my_app/application.ex"]
      end

      lib :my_app_web do
        # only eject the following files in `lib/my_app_web`
        only [
          "lib/my_app_web/endpoint.ex",
          "lib/my_app_web/router.ex",
          "lib/my_app_web/channels/user_socket.ex",
          "lib/my_app_web/views/error_view.ex",
          "lib/my_app_web/templates/layout/root.html.heex",
          "lib/my_app_web/templates/layout/app.html.heex",
          "lib/my_app_web/templates/layout/live.html.heex"
        ]
      end

      # always include these mix dependencies
      mix :credo
      mix :ex_doc
      mix :phoenix
      mix :phoenix_html
    end

    # if absinthe is included, also include absinthe_plug and dataloader
    mix :absinthe do
      mix_deps [:absinthe_plug, :dataloader]
    end

    lib :my_data_lib do
      # if my_data_lib is included, also include other_lib, faker, and norm
      lib_deps [:other_lib]
      mix_deps [:faker, :norm]

      # if my_data_lib is included, also eject these files
      file "priv/my_data_repo/**/*.exs", match_dot: true
      file "test/support/fixtures/my_data_lib/**/*.ex"
    end
  end
end

      


      
        Summary


  
    Callbacks
  


    
      
        app_lib_except(app)

      


        This callback works like the except/1 instruction for Lib Dependencies,
except that it applies to the lib folder of the ejected app itself.



    


    
      
        extra(app)

      


        A callback to add data to app.extra.



    


    
      
        target_path(path, app)

      


        Use this optional callback to change the path of files in the ejected
codebase.



    





  
    Functions
  


    
      
        always(deps)

      


        Inside of a deps do block, any Mix or Lib dependencies that should be
included in every ejected app should be wrapped in an always do block



    


    
      
        base_files(files)

      


        The base_files section is where you specify files outside of an ejected
app's lib/my_app and test/my_app directories which should always be
ejected.



    


    
      
        cp(path, opts \\ [])

      


        cp works exactly like file/2, except that no Code
Transformations are applied to the file.



    


    
      
        cp_r(path, opts \\ [])

      


        cp_r works like cp/2, but for a directory instead of a file. The
directory is copied as-is with File.cp_r!/3.



    


    
      
        deps(mix_and_lib_deps)

      


        Uniform automatically catalogs all Mix deps by looking into mix.exs to
discover all Mix deps.  It also catalogs all Lib deps by scanning the lib/
directory.



    


    
      
        except(paths)

      


        In the deps section of your Blueprint, you can specify that a Lib
Dependency excludes certain files.



    


    
      
        file(path, opts \\ [])

      


        In base_files and lib blocks, file is used to specify files that are
not in a lib/ directory which should be ejected in the app or along with
the lib.



    


    
      
        lib(name, list)

      


        Lib Dependencies are shared code
libraries in the lib directory of your project. Uniform scans lib/, so
you don't need to inform it about them.



    


    
      
        lib_deps(deps)

      


        Specify transitive Lib Dependencies of
other Lib Dependencies.



    


    
      
        mix(name, list)

      


        Uniform automatically catalogues the deps in mix.exs, so there are only two
scenarios where you need to list them in your Blueprint.



    


    
      
        mix_deps(deps)

      


        Specify transitive Mix Dependencies of
other Lib/Mix Dependencies.



    


    
      
        modify(path_or_regex, function)

      


        Modify the contents of one or more files during ejection.



    


    
      
        only(paths)

      


        In the deps section of your Blueprint, you can specify that a Lib
Dependency only includes certain files.



    


    
      
        template(path, opts \\ [])

      


        In base_files and lib blocks, template is used to specify EEx templates
that should be rendered and then ejected.



    





      


      
        Callbacks

        


  
    
      
      Link to this callback
    
    app_lib_except(app)


      
       
       View Source
     


      (optional)

  


  

      

          @callback app_lib_except(app :: Uniform.App.t()) :: [Path.t() | Regex.t()]


      


This callback works like the except/1 instruction for Lib Dependencies,
except that it applies to the lib folder of the ejected app itself.
When running mix uniform.eject my_app, any files in lib/my_app or
test/my_app which match the paths or regexes returned by app_lib_except
will not be ejected.
def app_lib_except(app) do
  ["lib/#{app.name.underscore}/hidden_file.ex"]
end

  



  
    
      
      Link to this callback
    
    extra(app)


      
       
       View Source
     


      (optional)

  


  

      

          @callback extra(app :: Uniform.App.t()) :: keyword()


      


A callback to add data to app.extra.
This callback exists for scenarios when you want to add an extra key to
many or all apps, and it can be programmatically determined from the
information in the app.
extra data that only applies to a single app usually belongs in the
Uniform Manifest.

  
    
  
  Example


You may want to set the theme based on the name of the ejectable app. Return
a keyword list with a theme key. It will be available via
app.extra[:theme] in modify/2, base_files/1, and
templates.
def extra(app) do
  theme =
    case app.name.underscore do
      "rebel_" <> _ -> :rebel
      "empire_" <> _ -> :empire
      _ -> raise "App name must start with rebel_ or empire_ to derive theme."
    end

  [theme: theme]
end
This prevents you from having to redundantly set
[
  extra: [theme: :rebel]
]
In every uniform.exs manifest.
uniform.exs has precedence
If uniform.exs is [extra: [key: :manifest]], app.extra[:key]
will (unsurprisingly) be :manifest in the extra/1 callback.
However, if extra/1 returns [key: :callback], app.extra[:key]
will still be :manifest in modify/2, base_files/1, and
templates.
In other words, uniform.exs "has precedence over" the extra/1
callback.


  



  
    
      
      Link to this callback
    
    target_path(path, app)


      
       
       View Source
     


      (optional)

  


  

      

          @callback target_path(path :: Path.t(), app :: Uniform.App.t()) :: Path.t()


      


Use this optional callback to change the path of files in the ejected
codebase.
It will be called for every ejected file, with the exception that cp_r/2
will only call it once for the entire directory.
If you don't want to modify the path, just return it.

  
    
  
  Example


You may want to place certain files in lib/ejected_app_web instead of
lib/ejected_app.  Let's say you have an is_web_file? function that
identifies whether the file belongs in the _web directory. target_path
might look like this:
def target_path(path, app) do
  if is_web_file?(path) do
    # modify the path to put it in `lib/some_app_web`
    String.replace(path, "lib/#{app.name.underscore}/", "lib/#{app.name.underscore}_web/")
  else
    path
  end
end

  


        

      

      
        Functions

        


  
    
      
      Link to this macro
    
    always(deps)


      
       
       View Source
     


      (macro)

  


  

Inside of a deps do block, any Mix or Lib dependencies that should be
included in every ejected app should be wrapped in an always do block:
deps do
  always do
    # always eject the contents of `lib/some_lib`
    lib :some_lib

    # always eject the some_mix Mix dependency
    mix :some_mix
  end
end

  



  
    
      
      Link to this macro
    
    base_files(files)


      
       
       View Source
     


      (macro)

  


  

The base_files section is where you specify files outside of an ejected
app's lib/my_app and test/my_app directories which should always be
ejected.
This section has access to an app variable which can
be used to build the instructions or conditionally include certain files with
if.
base_files do
  # interpolating the app name into a path dynamically
  cp "priv/static/images/#{app.name.underscore}_logo.png"

  # conditional instructions
  if deploys_to_fly_io?(app) do
    template "fly.toml"
  end
end
Note that if conditionals cannot be nested here.

  
    
  
  API Reference


	file/2 ejects a single file or list of files
	template/2 creates a new file on ejection from an EEx template
	cp/2 copies a file (like file/2) without running it through Code
Transformations. This is useful for binary
files such as images or executable.
	cp_r/2 copies an entire directory of files without Code
Transformations.


  
    
  
  Example


base_files do
  file ".credo.exs"
  file "config/**/*.exs"
  template "config/runtime.exs"
  cp "bin/some-executable", chmod: 0o555
  cp_r "assets"
end

  
    
  
  Files in lib


Typically, the base_files section only contains files that aren't in lib,
since files in lib/app_being_ejected and lib/required_lib_dependency are
ejected automatically.
# ❌ Don't do this
base_files do
  file "lib/my_lib/some_file.ex"
end

# ✅ Instead, do this (lib/my_app/uniform.exs)
[
  lib_deps: [:my_lib]
]

  
    
  
  Files outside lib but tied to Lib Dependencies


If a file or template should only be ejected in the case that a certain Lib
Dependency is included, we recommend placing that inside the deps section
instead of in base_files. (See Including files outside of
lib.)
# ❌ Don't do this
base_files do
  if depends_on?(app, :lib, :my_lib) do
    file "some_file"
  end
end

# ✅ Instead, do this
deps do
  lib :my_lib do
    file "some_file"
  end
end

  



    

  
    
      
      Link to this function
    
    cp(path, opts \\ [])


      
       
       View Source
     


  


  

cp works exactly like file/2, except that no Code
Transformations are applied to the file.
The file is copied as-is with File.cp!/3.

  
    
  
  Options


cp takes a chmod option, which sets the mode for the file after it's
copied. See the possible permission
options.

  
    
  
  Examples


base_files do
  # every ejected app will have bin/some-binary, with the ACL mode changed to 555
  cp "bin/some-binary", chmod: 0o555
end

deps do
  lib :pdf do
    # apps that include the pdf lib will also have bin/convert
    cp "bin/convert"
  end
end

  



    

  
    
      
      Link to this function
    
    cp_r(path, opts \\ [])


      
       
       View Source
     


  


  

cp_r works like cp/2, but for a directory instead of a file. The
directory is copied as-is with File.cp_r!/3.
None of the files are ran through Code
Transformations.
This is useful for directories that do not require modification, and contain
many files.
For example, the assets/node_modules directory in a Phoenix application
would take ages to copy with file "assets/node_modules/**/*". Instead, use
cp_r "assets/node_modules".

  
    
  
  Examples


base_files do
  cp_r "assets"
end

deps do
  lib :some_lib do
    cp_r "priv/files_for_some_lib"
  end
end

  



  
    
      
      Link to this macro
    
    deps(mix_and_lib_deps)


      
       
       View Source
     


      (macro)

  


  

      

          @spec deps(mix_and_lib_deps :: term()) :: term()


      


Uniform automatically catalogs all Mix deps by looking into mix.exs to
discover all Mix deps.  It also catalogs all Lib deps by scanning the lib/
directory.
If you need to configure anything about a Mix or Lib dep, such as other
dependencies that must be bundled along with it, use the deps block.
See mix/2, lib/2, and always/1 for more details.

  
    
  
  Example


deps do
  always do
    lib :my_app do
      only ["lib/my_app/application.ex"]
    end

    mix :phoenix
  end

  mix :absinthe do
    mix_deps [:absinthe_plug, :dataloader]
  end

  lib :my_custom_aws_lib do
    lib_deps [:my_utilities_lib]
    mix_deps [:ex_aws, :ex_aws_ec2]
  end
end

  



  
    
      
      Link to this function
    
    except(paths)


      
       
       View Source
     


  


  

In the deps section of your Blueprint, you can specify that a Lib
Dependency excludes certain files.
This works much like the except option that can be given when importing
functions with import/2.
In the example below, for any app that depends on :aws, every file in
lib/aws and test/aws will be ejected except for lib/aws/hidden_file.ex.
deps do
  lib :aws do
    except ["lib/aws/hidden_file.ex"]
  end
end

  



    

  
    
      
      Link to this function
    
    file(path, opts \\ [])


      
       
       View Source
     


  


  

In base_files and lib blocks, file is used to specify files that are
not in a lib/ directory which should be ejected in the app or along with
the lib.

  
    
  
  Options


	chmod – Sets the mode for the given file after it's ejected. See the
possible permission options.
	match_dot – Forwarded to Path.wildcard/2. See "Wildcard Globs" below.


  
    
  
  Glob Expressions


You can use a glob expression with wildcard characters
to target multiple files. (See "Examples" below.)
This is possible because Uniform internally forwards the path and opts to
Path.wildcard/2 to determine which files to eject.
Note that the * and ? "wildcard characters" will not match files starting
with a dot (.) unless you provide match_dot: true.

  
    
  
  Examples


base_files do
  file "assets/js/app.js"
  file "some/file", chmod: 0o777

  # glob targeting .exs files in config/
  file "config/**/*.exs"

  # glob targeting .exs files in priv/repo/ – including .formatter.exs
  file "priv/repo/**/*.exs", match_dot: true
end

deps do
  lib :aws do
    # for every app that includes the aws lib dependency,
    # some_aws_fixture.xml will also be included
    file "test/support/fixtures/some_aws_fixture.xml"
  end
end

  



  
    
      
      Link to this macro
    
    lib(name, list)


      
       
       View Source
     


      (macro)

  


  

Lib Dependencies are shared code
libraries in the lib directory of your project. Uniform scans lib/, so
you don't need to inform it about them.
However, there are four scenarios where you do need to list them in your
Blueprint.

  
    
  
  1. Specifying which Lib Dependencies should always be ejected


deps do
  always do
    # every app will have lib/utilities
    lib :utilities
  end
end
See always/1.

  
    
  
  2. When a Lib Dependency has other Lib or Mix Dependencies


deps do
  # if `lib/auth` is included, tesla and `lib/utils` will be included
  lib :auth do
    mix_deps [:tesla]
    lib_deps [:utils]
  end
end
See mix_deps/1 and lib_deps/1.

  
    
  
  3. Including files outside of lib


Some libraries require other files outside of lib/that_library.
For example:
deps do
  # when `lib/my_data_source is included...
  lib :my_data_source do
    # files in priv/my_data_source_repo will be included
    file "priv/my_data_source_repo/**", match_dot: true

    # .ex files in `test/support/my_data_source` will be included
    file "test/support/my_data_source/**/*.ex"

    # `priv/my_data_source_file` will be included from a template
    template "priv/my_data_source_file"
  end
end
See file/2, template/2, cp/1, and cp_r/1.

  
    
  
  4. When certain files should be excluded from a Lib Dependency upon ejection


deps do
  always do
    # every app will have lib/mix, but only `some_task.ex` will be ejected
    lib :mix do
      only ["lib/mix/tasks/some_task.ex"]
    end
  end

  # `some_file.ex` will be omitted from `lib/auth` in ejected codebases
  lib :auth do
    except ["lib/auth/some_file.ex"]
  end
end
See only/1 and except/1.

  



  
    
      
      Link to this macro
    
    lib_deps(deps)


      
       
       View Source
     


      (macro)

  


  

Specify transitive Lib Dependencies of
other Lib Dependencies.
Libraries listed with lib_deps will be included in the ejected codebase any
time the "parent" dependency is included.

  
    
  
  Examples


deps do
  # `lib/core_auth` will never be ejected without `lib/oauth_lib`
  lib :core_auth do
    lib_deps [:oauth_lib]
  end
end

  



  
    
      
      Link to this macro
    
    mix(name, list)


      
       
       View Source
     


      (macro)

  


  

Uniform automatically catalogues the deps in mix.exs, so there are only two
scenarios where you need to list them in your Blueprint.

  
    
  
  1. Specifying deps that should always be ejected.


deps do
  always do
    # every app will have credo and ex_doc
    mix :credo
    mix :ex_doc
  end
end
See always/1.

  
    
  
  2. When a Mix Dependency has other Mix Dependencies.


deps do
  mix :oban do
    # any app that is ejected with oban will also have oban_pro and oban_web
    mix_deps [:oban_pro, :oban_web]
  end
end
See mix_deps/1.

  



  
    
      
      Link to this macro
    
    mix_deps(deps)


      
       
       View Source
     


      (macro)

  


  

Specify transitive Mix Dependencies of
other Lib/Mix Dependencies.
Dependencies listed with mix_deps will be included in the ejected mix.exs
any time the "parent" dependency is included.

  
    
  
  Examples


deps do
  # if absinthe is included, absinthe_plug and dataloader will be included
  mix :absinthe do
    mix_deps [:absinthe_plug, :dataloader]
  end

  # when `lib/ui_components` is included, surface will be included
  lib :ui_components do
    mix_deps [:surface]
  end
end

  



  
    
      
      Link to this macro
    
    modify(path_or_regex, function)


      
       
       View Source
     


      (macro)

  


  

      

          @spec modify(
  pattern :: Path.t() | Regex.t(),
  function ::
    (file :: String.t(), Uniform.App.t() -> String.t())
    | (file :: String.t() -> String.t())
) :: term()


      


Modify the contents of one or more files during ejection.
Takes a transformation function which returns the new file contents as a
string.
The first argument is either the relative path of a file in your Base
Project or a Regex.
# modify a specific test
modify "tests/path/to/specific_test.exs", fn file -> ... end

# modify all `_test.exs` files
modify ~r/_test.exs$/, fn file -> ... end
The second argument is either a function capture
modify ~r/.+_test.exs/, &MyApp.Modify.modify_tests/1
modify ~r/.+_test.exs/, &MyApp.Modify.modify_tests/2
Or an anonymous function
modify ~r/.+_test.exs/, fn file ->
  # ...
end

modify ~r/.+_test.exs/, fn file, app ->
  # ...
end
If the function is 1-arity, it will receive the file contents. If it's
2-arity, it will receive the file contents and the Uniform.App struct.

  
    
  
  Examples


modify "config/config.exs", file do
  file <>
    ~S'''
    if config_env() in [:dev, :test] do
      import_config "#{config_env()}.exs"
    end
    '''
end

modify ~r/.+_worker.ex/, fn file, app ->
  String.replace(file, "SOME_VALUE_PER_APP", app.extra[:some_value])
end

  



  
    
      
      Link to this function
    
    only(paths)


      
       
       View Source
     


  


  

In the deps section of your Blueprint, you can specify that a Lib
Dependency only includes certain files.
These work much like the only option that can be given when importing
functions with import/2.
In the example below, for any app that depends on :gcp, only
lib/gcp/necessary_file.ex will be ejected. Nothing else from lib/gcp or
test/gcp will be ejected.
deps do
  lib :gcp do
    # NOTHING in lib/gcp or test/gcp will be included except these:
    only ["lib/gcp/necessary_file.ex"]
  end
end

  



    

  
    
      
      Link to this function
    
    template(path, opts \\ [])


      
       
       View Source
     


  


  

In base_files and lib blocks, template is used to specify EEx templates
that should be rendered and then ejected.

  
    
  
  Template Directory and Destination Path


Uniform templates use a "convention over configuration" model that works like
this:
	At the top of your Blueprint module, you specify a template directory
like this:
 use Uniform, templates: "lib/uniform/templates"

	Templates must be placed in this directory at the relative path that they
should be placed in, in the ejected directory.

	Templates must have the destination filename, appended with .eex.


Companion guide
Consult Building files from EEx
templates for a more detailed look
at constructing and effectively using templates for ejection.


  
    
  
  Options


template takes a chmod option, which sets the mode for the rendered
file after it's ejected. See the possible permission
options.

  
    
  
  Examples


use Uniform, templates: "priv/uniform-templates"

base_files do
  # priv/uniform-templates/config/runtime.exs.eex will be rendered, and the
  # result will be placed in `config/runtime.exs` in every ejected app
  template "config/runtime.exs"
end

deps do
  lib :datadog do
    # for every app that includes `lib/datadog`,
    # priv/uniform-templates/datadog/prerun.sh.eex will be rendered, and
    # the result will be placed in `datadog/prerun.sh`
    template "datadog/prerun.sh", chmod: 0o555
  end
end

  


        

      



  

    
Uniform.Modifiers 
    



      
Utilities for building code transformations with
Uniform.Blueprint.modify/2.

      


      
        Summary


  
    Functions
  


    
      
        eject_fences(file_contents, app, prefix, suffix \\ nil)

      


        Apply Eject Fences to extra
languages with this function.



    





      


      
        Functions

        


    

  
    
      
      Link to this function
    
    eject_fences(file_contents, app, prefix, suffix \\ nil)


      
       
       View Source
     


  


  

      

          @spec eject_fences(
  String.t(),
  Uniform.App.t(),
  prefix :: String.t(),
  suffix :: String.t() | nil
) :: String.t()


      


Apply Eject Fences to extra
languages with this function.
Uniform already applies Eject Fences to Elixir (.ex/.exs) and JavaScript
(.js/.jsx/.ts/.tsx) files automatically.
use Uniform.Blueprint imports eject_fences.

  
    
  
  Examples


# code fences for SQL files
modify ~r/\.sql$/, &eject_fences(&1, &2, "--")

# code fences for Rust files
modify ~r/\.rs$/, fn file, app ->
  eject_fences(file, app, "//")
end
Note that the prefix and suffix are placed directly into a Regex using
Regex.compile!/2.  Therefore, if the prefix and suffix include special
Regex characters such as *, they must be escaped appropriately.
For example, the following would process CSS /* comments */.
# code fences for CSS files
modify ~r/\.css$/, fn file, app ->
  eject_fences(file, app, "/\\*", "\\*/")
end

  


        

      



  

    
mix uniform.eject 
    



      
Ejects an Ejectable App to a
standalone code repository.

  
    
  
  Usage


$ mix uniform.eject trillo
$ mix uniform.eject tweeter --confirm
$ mix uniform.eject hatmail --confirm --destination ../../new/dir


  
    
  
  Command line options


	--destination – output directory for the ejected code. Read the
Configuration section of the Getting Started
guide to understand how the
destination is chosen if this option is omitted.
	--confirm – affirm "yes" to the prompt asking you whether you want to eject.


  
    
  
  Which files get ejected


When you run mix uniform.eject my_app, these four rules determine which files
are copied.
	A few files
common to Elixir projects are copied.
	All files in the Blueprint's
base_files section are copied.
	All files in lib/my_app and test/my_app are copied.
	For every Lib Dependency of my_app:	All files in lib/my_lib_dep and test/my_lib_dep are copied.
	All associated
files
tied to the Lib Dependency are copied.



If you need to apply exceptions to these rules, you can use these tools.
	Files in (lib|test)/my_app (rule 3) are subject to the
lib_app_except callback.
	Lib Dependency files (rule 4) are subject to
only and
except instructions.



  
    
  
  Ejection step by step


When you eject an app by running mix uniform.eject my_app, the following happens:
	The destination directory is created if it doesn't exist.
	All files and directories in the destination are deleted, except for
.git, _build, deps, and anything in the Blueprint's
@preserve.
	All files required by the app are copied to the destination. (See Which
files get ejected.)
	For each file copied, a set of
transformations are applied to the file
contents – except for files specified with cp and cp_r.
	mix deps.clean --unlock --unused is ran on the ejected codebase.
	mix format is ran on the ejected codebase.

In step 2, .git is kept to preserve the Git repository and history. deps
is kept to avoid having to download all dependencies after ejection. _build
is kept to avoid having to recompile the entire project after ejection.
mix deps.clean --unlock --unused removes unused Mix Dependencies from
mix.lock in the ejected codebase. This includes deps removed from
mix.exs as well as transitive
dependencies of those deps.
mix format tidies up things like chains of newlines that may appear from
applying Eject Fences. It also
prevents you from having to think about code formatting in
modify.

      





  

    
mix uniform.ejectable_apps 
    



      
Outputs the name of all Ejectable Apps
to stdout.
This task is useful for building a CI pipeline that automatically commits
updates to each ejected code repository. See the Auto-updating ejected
codebases guide for more information.

  
    
  
  Usage


$ mix uniform.ejectable_apps
tweeter
trillo
hatmail


      





  

    
mix uniform.gen.app 
    



      
Generates an uniform.exs file so that the given app can be ejected with mix uniform.eject.

  
    
  
  Usage


$ mix uniform.gen.app some_app_name
Created lib/some_app_name/uniform.exs

Running that command will create lib/some_app_name/uniform.exs, which will
allow you to run
$ mix uniform.eject some_app_name


      





  

    
mix uniform.init 
    



      
Initializes a Base Project repository
with the bare minimum setup required for using Uniform.
	Adds a Blueprint module
	Adds required configuration to config/config.exs

The remainder of installation steps are listed in the Getting
Started guide.

  
    
  
  Usage


$ mix uniform.init
Created lib/base_app/uniform/blueprint.ex
Added configuration in config/config.exs

